
 

Declanşatori în PL/SQL 
 

Un declanşator (trigger) este un bloc PL/SQL  sau apelul (CALL) unei 

proceduri PL/SQL, care se execută automat ori de câte ori are loc un anumit 

eveniment „declanşator“. Evenimentul poate consta din: 

 modificarea unui tabel sau a unei vizualizări,  

 acţiuni sistem 

 anumite acţiuni utilizator. 

Blocul PL/SQL poate fi asociat unui tabel, unei vizualizări, unei scheme 

sau unei baze de date.  

La fel ca şi pachetele, declanşatorii nu pot fi locali unui bloc sau unui 

pachet, ei trebuie depuşi ca obiecte independente în baza de date. 

Folosirea declanşatorilor garantează faptul că atunci când o anumită 

operaţie este efectuată, automat sunt executate nişte acţiuni asociate. Evident, nu 

trebuie introduşi declanşatori care ar putea să substituie funcţionalităţi oferite deja 

de sistem. De exemplu, nu are sens să fie definiţi declanşatori care să 

implementeze regulile de integritate ce pot fi definite, mai simplu, prin 

constrângeri declarative.  

Tipuri de declanşatori 

Declanşatorii pot fi: 

 la nivel de bază de date (database triggers); 

 la nivel de aplicaţie (application triggers).  

Declanşatorii bază de date se execută automat ori de câte ori are loc: 

 o acţiune (comandă LMD) asupra datelor unui tabel; 

 o acţiune (comandă LMD) asupra datelor unei vizualizări; 

 o comandă LDD (CREATE, ALTER, DROP) referitoare la anumite 

obiecte ale schemei sau ale bazei; 

 un eveniment sistem (SHUTDOWN, STARTUP); 

 o acţiune a utilizatorului (LOGON, LOGOFF); 

 o eroare (SERVERERROR, SUSPEND).  

Declanşatorii bază de date sunt de trei tipuri:  

 declanşatori LMD – activaţi de comenzi LMD (INSERT, UPDATE sau 

DELETE) executate asupra unui tabel al bazei de date; 

 declanşatori INSTEAD OF – activaţi de comenzi LMD executate asupra 

unei vizualizări (relaţionale sau obiect); 

 declanşatori sistem – activaţi de un eveniment sistem (oprirea sau 

pornirea bazei), de comenzi LDD (CREATE, ALTER, DROP), de 

conectarea (deconectarea) unui utilizator. Ei sunt definiţi la nivel de 

schemă sau la nivel de bază de date.  

Declanşatorii asociaţi unui tabel (stocaţi în baza de date) vor acţiona 

indiferent de aplicaţia care a efectuat operaţia LMD. Dacă operaţia LMD se referă 

la o vizualizare, declanşatorul INSTEAD OF defineşte acţiunile care vor avea loc, 

iar dacă aceste acţiuni includ comenzi LMD referitoare la tabele, atunci 

declanşatorii asociaţi acestor tabele sunt şi ei, la rândul lor, activaţi.  

Dacă declanşatorii sunt asociaţi unei baze de date, ei se declanşează pentru 

fiecare eveniment, pentru toţi utilizatorii. Dacă declanşatorii sunt asociaţi unei 

scheme sau unui tabel, ei se declanşează numai dacă evenimentul declanşator 

implică acea schemă sau acel tabel. Un declanşator se poate referi la un singur 

tabel sau la o singură vizualizare. 

Declanşatorii aplicaţie se execută implicit ori de câte ori apare un 

eveniment particular într-o aplicaţie (de exemplu, o aplicaţie dezvoltată cu 

Developer Suite). Form Builder utilizează frecvent acest tip de declanşatori (form 

builder triggers). Ei pot fi declanşaţi prin apăsarea unui buton, prin navigarea pe 

un câmp etc. În acest capitol se va face referinţă doar la declanşatorii bază de 

date. 

Atunci când un pachet sau un subprogram este depus în dicţionarul datelor, 

alături de codul sursă este depus şi p-codul compilat. În mod similar se întâmplă 

şi pentru declanşatori. Prin urmare, un declanşator poate fi apelat fără 

recompilare. Declanşatorii pot fi invalidaţi în aceeeaşi manieră ca pachetele şi 

subprogramele. Dacă declanşatorul este invalidat, el va fi recompilat la 

următoarea activare. 

Crearea declanşatorilor LMD 

Declanşatorii LMD sunt creaţi folosind comanda CREATE TRIGGER. 

Numele declanşatorului trebuie să fie unic printre numele declanşatorilor 

din cadrul aceleaşi scheme, dar poate să coincidă cu numele altor obiecte ale 

acesteia (de exemplu, tabele, vizualizări sau proceduri).  

La crearea unui declanşator este obligatorie una dintre opţiunile BEFORE 

sau AFTER, prin care se precizează momentul în care este executat corpul 

declanşatorului. Acesta nu poate depăşi 32KB. 

CREATE [OR REPLACE] TRIGGER [schema.]nume_declanşator 

  {BEFORE | AFTER} 

  {DELETE | INSERT | UPDATE [OF coloana[, coloana …] ] } 

  [OR {DELETE | INSERT | UPDATE [OF coloana[, coloana …] ] …} 

  ON [schema.]nume_tabel 

  [REFERENCING {OLD [AS] vechi NEW [AS] nou 

                               | NEW [AS] nou OLD [AS] vechi } ] 

  [FOR EACH ROW] 

  [WHEN (condiţie) ] 

  corp_declanşator (bloc PL/SQL sau apelul unei proceduri); 

Până la versiunea Oracle8i, corpul unui declanşator trebuia să fie un bloc 

PL/SQL. În ultimele versiuni, corpul poate consta doar dintr-o singură comandă 

CALL. Procedura apelată poate fi un subprogram PL/SQL stocat, o rutină C sau o 

metodă Java. În acest caz, CALL nu poate conţine clauza INTO care este 

specifică funcţiilor, iar pentru a referi coloanele tabelului asociat declanşatorului, 

acestea trebuie prefixate de atributele :NEW sau :OLD. De asemenea, în expresia 

parametrilor nu pot să apară variabile bind. 

Declararea unui declanşator trebuie să cuprindă tipul comenzii SQL care 

duce la executarea declanşatorului şi tabelul asociat acestuia. În ceea ce priveşte 

tipul comenzii SQL care va duce la executarea declaşatorului, sunt incluse 

următoarele tipuri de opţiuni: DELETE, INSERT, UPDATE sau o combinare a 

acestora cu operatorul logic OR. Cel puţin una dintre opţiuni este obligatorie.  

În declararea declanşatorului este specificat tabelul asupra căruia va fi 

executat declanşatorul. Oracle9i admite tablouri imbricate. Dacă declanşatorul 

este de tip UPDATE, atunci pot fi enumerate coloanele pentru care acesta se va 

executa.  

În corpul fiecărui declanşator pot fi cunoscute valorile coloanelor atât 

înainte de modificarea unei linii, cât şi după modificarea acesteia. Valoarea unei 

coloane înainte de modificare este referită prin atributul OLD, iar după 

modificare, prin atributul NEW. Prin intermediul clauzei opţionale 

REFERENCING din sintaxa comenzii de creare a declanşatorilor, atributele NEW 

şi OLD pot fi redenumite. In interiorul blocului PL/SQL, coloanele prefixate prin 

OLD sau NEW sunt considerate variabile externe, deci trebuie prefixate cu ":". 

Un declanşator poate activa alt declanşator, iar acesta la rândul său poate 

activa alt declanşator etc. Această situaţie (declanşatori în cascadă) poate avea 

însă efecte imprevizibile. Sistemul Oracle permite maximum 32 declanşatori în 

cascadă. Numărul acestora poate fi limitat (utilizând parametrul de iniţializare 

OPEN_CURSORS), deoarece pentru fiecare execuţie a unui declanşator trebuie 

deschis un nou cursor. 

Declanşatorii la nivel de baze de date pot fi de două feluri:  

 la nivel de instrucţiune (statement level trigger); 

 la nivel de linie (row level trigger).  

Declanşatori la nivel de instrucţiune 

Declanşatorii la nivel instrucţiune sunt executaţi o singură dată pentru 

instrucţiunea declanşatoare, indiferent de numărul de linii afectate (chiar dacă nici 

o linie nu este afectată). Un declanşator la nivel de instrucţiune este util dacă 

acţiunea declanşatorului nu depinde de informaţiile din liniile afectate. 

Exemplu: 

Programul de lucru la administraţia muzeului este de luni până vineri, în 

intervalul (8:00 a.m. - 10:00 p.m.). Să se construiască un declanşator la nivel de 

instrucţiune care împiedică orice activitate asupra unui tabel al bazei de date, în 

afara acestui program. 

CREATE OR REPLACE PROCEDURE verifica IS 

BEGIN 

  IF ((TO_CHAR(SYSDATE,'D') BETWEEN 2 AND 6) 

    AND 

      TO_DATE(TO_CHAR(SYSDATE,'hh24:mi'), 'hh24:mi') 

        NOT BETWEEN TO_DATE('08:00','hh24:mi') 

                AND TO_DATE('22:00','hh24:mi')) 

    THEN 

      RAISE_APPLICATION_ERROR (-27733, 'nu puteti reactualiza 

         acest tabel deoarece sunteti in afara programului'); 

  END IF; 

END verifica; 

/ 

CREATE OR REPLACE TRIGGER BIUD_tabel1 

  BEFORE INSERT OR UPDATE OR DELETE ON tabel1 

BEGIN 

  verifica; 

END; 

/ 

Declanşatori la nivel de linie 

Declanşatorii la nivel de linie sunt creaţi cu opţiunea FOR EACH ROW. În 

acest caz, declanşatorul este executat pentru fiecare linie din tabelul afectat, iar 

dacă evenimentul declanşator nu afectează nici o linie, atunci declanşatorul nu 

este executat. Dacă opţiunea FOR EACH ROW nu este inclusă, declanşatorul este 

considerat implicit la nivel de instrucţiune. 

Declanşatorii la nivel linie nu sunt performanţi dacă se fac frecvent 

reactualizări pe tabele foarte mari. 

Restricţiile declanşatorilor pot fi incluse prin specificarea unei expresii 

booleene în clauza WHEN. Acestă expresie este evaluată pentru fiecare linie 

afectată de către declanşator. Declanşatorul este executat pentru o linie, doar 

dacă expresia este adevărată pentru acea linie. Clauza WHEN este validă doar 

pentru declanşatori la nivel de linie. 

Exemplu: 

Să se implementeze cu ajutorul unui declanşator constrângerea că valorile 

operelor de artă nu pot fi reduse (trei variante). 

Varianta 1: 

CREATE OR REPLACE TRIGGER verifica_valoare 

  BEFORE UPDATE OF valoare ON opera 

  FOR EACH ROW 

  WHEN (NEW.valoare < OLD.valoare) 

BEGIN 

  RAISE_APPLICATION_ERROR (-20222, 'valoarea unei opere de 

                           arta nu poate fi micsorata'); 

END; 

Varianta 2: 

CREATE OR REPLACE TRIGGER verifica_valoare 

  BEFORE UPDATE OF valoare ON opera 

  FOR EACH ROW 

BEGIN 

  IF (:NEW.valoare < :OLD.valoare) THEN  

     RAISE_APPLICATION_ERROR (-20222, 'valoarea unei opere de 
                             arta nu poate fi micsorata'); 

  END IF; 

END; 

Varianta 3: 

CREATE OR REPLACE TRIGGER verifica_valoare  

  BEFORE UPDATE OF valoare ON opera 

  FOR EACH ROW 

  WHEN (NEW.valoare < OLD.valoare) 

  CALL procedura -- care va face actiunea RAISE … 

/ 

Accesul la vechile şi noile valori ale coloanelor liniei curente, afectată de 

evenimentul declanşator, se face prin: OLD.nume_coloană (vechea valoare), 

respectiv prin NEW.nume_coloană (noua valoare). În cazul celor trei comenzi 

LMD, aceste valori devin: 

INSERT  : NEW.nume_coloană  noua valoare  

(: OLD.nume_coloană  NULL ); 

UPDATE  : NEW.nume_coloană  noua valoare 

: OLD.nume_coloană  vechea valoare; 

DELETE  (: NEW.nume_coloană  NULL ) 

: OLD.nume_coloană  vechea valoare. 

Exemplu: 

Se presupune că pentru fiecare galerie există două câmpuri (min_valoare şi 

max_valoare) în care se reţin limitele minime şi maxime ale valorile operelor din 

galeria respectivă. Să se implementeze cu ajutorul unui declanşator constrângerea 

că, dacă aceste limite s-ar modifica, valoarea oricarei opere de artă trebuie să 

ramană cuprinsă între noile limite.  

CREATE OR REPLACE TRIGGER verifica_limite 

  BEFORE UPDATE OF min_valoare, max_valoare ON galerie 

  FOR EACH ROW 

DECLARE 

  v_min_val  opera.valoare%TYPE; 

  v_max_val  opera.valoare%TYPE; 

  e_invalid  EXCEPTION; 

BEGIN 

  SELECT MIN(valoare), MAX(valoare) 

  INTO   v_min_val, v_max_val 

  FROM   opera 

  WHERE  cod_galerie = :NEW.cod_galerie; 

  IF (v_min_val < :NEW.min_valoare) OR  

     (v_max_val > :NEW.max_valoare) THEN 

     RAISE e_invalid; 

  END IF; 

EXCEPTION 

  WHEN e_invalid THEN 

    RAISE_APPLICATION_ERROR (-20567, 'Exista opere de 

    arta ale caror valori sunt in afara domeniului 

    permis'); 

END verifica_limite; 

/ 

Ordinea de execuţie a declanşatorilor 

PL/SQL permite definirea a 12 tipuri de declanşatori care sunt obţinuţi prin 

combinarea proprietăţii de moment (timp) al declanşării (BEFORE, AFTER), cu 

proprietatea nivelului la care acţionează (nivel linie, nivel intrucţiune) şi cu tipul 

operaţiei ataşate declanşatorului (INSERT, UPDATE, DELETE).  

De exemplu, BEFORE INSERT acţionează o singură dată, înaintea 

executării unei instrucţiuni INSERT, iar BEFORE INSERT FOR EACH ROW 

acţionează înainte de inserarea fiecărei noi înregistrări. 

Declanşatorii sunt activaţi când este executată o comandă LMD. La 

apariţia unei astfel de comenzi se execută câteva acţiuni care vor fi descrise în 

continuare. 

1. Se execută declanşatorii la nivel de instrucţiune BEFORE. 

2. Pentru fiecare linie afectată de comanda LMD: 

2.1. se execută declanşatorii la nivel de linie BEFORE; 

2.2. se blochează şi se modifică linia afectată (se execută comanda LMD), se 

verifică constrângerile de integritate (blocarea rămâne valabilă până în 

momentul în care tranzacţia este permanentizată); 

2.3. se execută declanşatorii la nivel de linie AFTER. 

3. Se execută declanşatorii la nivel de instrucţiune AFTER. 

Începând cu versiunea Oracle8i algoritmul anterior se schimbă, în sensul 

că verificarea constrângerii referenţiale este amânată după executarea 

declanşatorului la nivel linie. 

Obsevaţii: 

 În expresia clauzei WHEN nu pot fi incluse funcţii definite de utilizator 

sau subcereri SQL. 

 În clauza ON poate fi specificat un singur tabel sau o singură 

vizualizare. 

 În interiorul blocului PL/SQL, coloanele tabelului prefixate cu OLD sau 

NEW sunt considerate variabile externe şi deci, trebuie precedate de 

caracterul „:“.  

 Condiţia de la clauza WHEN poate conţine coloane prefixate cu OLD 

sau NEW, dar în acest caz, acestea nu trebuie precedate de „:“.  

 Declanşatorii bază de date pot fi definiţi numai pe tabele (excepţie, 

declanşatorul INSTEAD OF care este definit pe o vizualizare). Totuşi, 

dacă o comandă LMD este aplicată unei vizualizări, pot fi activaţi 

declanşatorii asociaţi tabelelor care definesc vizualizarea. 

 Corpul unui declanşator nu poate conţine o interogare sau o 

reactualizare a unui tabel aflat în plin proces de modificare, pe timpul 

acţiunii declanşatorului (mutating table).  

 Blocul PL/SQL care descrie acţiunea declanşatorului nu poate conţine 

comenzi pentru gestiunea tranzacţiilor (COMMIT, ROLLBACK, 

SAVEPOINT). Controlul tranzacţiilor este permis, însă, în procedurile 

stocate. Dacă un declanşator apelează o procedură stocată care execută 

o comandă referitoare la controlul tranzacţiilor, atunci va apărea o 

eroare la execuţie şi tranzacţia va fi anulată. 

 Comenzile LDD nu pot să apară decât în declanşatorii sistem. 

 Corpul declanşatorului poate să conţină comenzi LMD. 

 În corpul declanşatorului pot fi referite şi utilizate coloane LOB, dar nu 

pot fi modificate valorile acestora.  

 Nu este indicată crearea declanşatorilor recursivi. 

 În corpul declanşatorului se pot insera date în coloanele de tip LONG şi 

LONGRAW, dar nu pot fi declarate variabile de acest tip. 

 Dacă un tabel este suprimat (se şterge din dicţionarul datelor), automat 

sunt distruşi toţi declanşatorii asociaţi tabelului. 

 Este necesară limitarea dimensiunii unui declanşator. Dacă acesta 

solicită mai mult de 60 linii de cod, atunci este preferabil ca o parte din 

cod să fie inclusă într-o procedură stocată şi aceasta să fie apelată din 

corpul declanşatorului. 

Sunt două diferenţe esenţiale între declanşatori şi procedurile stocate:  

 declanşatorii se invocă implicit, iar procedurile explicit;  

 instrucţiunile LCD (COMMIT, ROLLBACK, SAVEPOINT) nu sunt 

permise în corpul unui declanşator. 

Predicate condiţionale 

În interiorul unui declanşator care poate fi executat pentru diferite tipuri de 

instrucţiuni LMD se pot folosi trei funcţii booleene prin care se stabileşte tipul 

operaţiei executate. Aceste predicate condiţionale (furnizate de pachetul standard 

DBMS_STANDARD) sunt INSERTING, UPDATING şi DELETING. 

Funcţiile booleene nu solicită prefixarea cu numele pachetului şi determină 

tipul operaţiei (INSERT, DELETE, UPDATE). De exemplu, predicatul 

INSERTING ia valoarea TRUE dacă instrucţiunea declanşatoare este INSERT. 

Similar sunt definite predicatele UPDATING şi DELETING. Utilizând aceste 

predicate, în corpul declanşatorului se pot executa secvenţe de instrucţiuni 

diferite, în funcţie de tipul operaţiei LMD. 

În cazul în care corpul declanşatorului este un bloc PL/SQL complet (nu o 

comandă CALL), pot fi utilizate atât predicatele INSERTING, UPDATING, 

DELETING, cât şi identificatorii :OLD, :NEW, :PARENT. 

Exemplu: 

Se presupune că în tabelul galerie se păstrează (într-o coloană numită 

total_val) valoarea totală a operelor de artă expuse în galeria respectivă. 

UPDATE galerie 

SET    total_val =  

           (SELECT SUM(valoare) 

            FROM   opera 

            WHERE opera.cod_galerie = galerie.cod_galerie); 

Reactualizarea acestui câmp poate fi implementată cu ajutorul unui 

declanşator în următoarea manieră: 

CREATE OR REPLACE PROCEDURE creste 

          (v_cod_galerie  IN galerie.cod_galerie%TYPE, 

           v_val         IN galerie.total_val%TYPE) AS 

BEGIN 

  UPDATE  galerie 

  SET     total_val = NVL (total_val, 0) + v_val 

  WHERE   cod_galerie = v_cod_galerie; 

END creste; 

/ 

CREATE OR REPLACE TRIGGER calcul_val 

  AFTER INSERT OR DELETE OR UPDATE OF valoare ON opera 

  FOR EACH ROW 

BEGIN 

  IF DELETING THEN  

    creste (:OLD.cod_galerie, -1*:OLD.valoare); 

  ELSIF UPDATING THEN  

    creste (:NEW.cod_galerie, :NEW.valoare - :OLD.valoare); 

  ELSE /* inserting */  

    creste (:NEW.cod_galerie, :NEW.valoare); 

  END IF; 

END; 

/ 

Declanşatori INSTEAD OF 

PL/SQL permite definirea unui nou tip de declanşator, numit INSTEAD OF, 

care oferă o modalitate de actualizare a vizualizărilor obiect şi a celor relaţionale. 

Sintaxa acestui tip de declanşator este similară celei pentru declanşatori 

LMD, cu două excepţii: 

 clauza {BEFORE | AFTER} este înlocuită prin INSTEAD OF; 

 clauza ON [schema.]nume_tabel este înlocuită printr-una din clauzele 

ON [schema.]nume_view sau ON NESTED TABLE (nume_coloană) 

OF [schema.]nume_view. 

Declanşatorul INSTEAD OF permite reactualizarea unei vizualizări prin 

comenzi LMD. O astfel de modificare nu poate fi realizată în altă manieră, din 

cauza regulilor stricte existente pentru reactualizarea vizualizărilor. Declanşatorii 

de tip INSTEAD OF sunt necesari, deoarece vizualizarea pe care este definit 

declanşatorul poate, de exemplu, să se refere la join-ul unor tabele, şi în acest 

caz, nu sunt actualizabile toate legăturile.  

O vizualizare nu poate fi modificată prin comenzi LMD dacă vizualizarea 

conţine operatori pe mulţimi, funcţii grup, clauzele GROUP BY, CONNECT BY, 

START WITH, operatorul DISTINCT sau join-uri.  

Declanşatorul INSTEAD OF este utilizat pentru a executa operaţii LMD 

direct pe tabelele de bază ale vizualizării. De fapt, se scriu comenzi LMD relative 

la o vizualizare, iar declanşatorul, în locul operaţiei originale, va opera pe tabelele 

de bază. 

De asemenea, acest tip de declanşator poate fi definit asupra vizualizărilor 

ce au drept câmpuri tablouri imbricate, declanşatorul furnizând o modalitate de 

reactualizare a elementelor tabloului imbricat.  

În acest caz, el se declanşează doar în cazul în care comenzile LMD 

operează asupra tabloului imbricat (numai când elementele tabloului imbricat sunt 

modificate folosind clauzele THE() sau TABLE()) şi nu atunci când comanda 

LMD operează doar asupra vizualizării. Declanşatorul permite accesarea liniei 

„părinte“ ce conţine tabloul imbricat modificat. 

Observaţii: 

 Spre deosebire de declanşatorii BEFORE sau AFTER, declanşatorii 

INSTEAD OF se execută în locul instrucţiunii LMD (INSERT, 

UPDATE, DELETE) specificate.  

 Opţiunea UPDATE OF nu este permisă pentru acest tip de declanşator. 

 Declanşatorii INSTEAD OF se definesc pentru o vizualizare, nu pentru 

un tabel.  

 Declanşatorii INSTEAD OF acţionează implicit la nivel de linie. 

 Dacă declanşatorul este definit pentru tablouri imbricate, atributele 

:OLD şi :NEW se referă la liniile tabloului imbricat, iar pentru a referi 

linia curentă din tabloul „părinte“ s-a introdus atributul :PARENT. 

Exemplu: 

Se consideră nou_opera, respectiv nou_artist, copii ale tabelelor opera, 

respectiv artist şi vi_op_ar o vizualizare definită prin compunerea naturală a 

celor două tabele. Se presupune că pentru fiecare artist există un câmp (sum_val) 

ce reprezintă valoarea totală a operelor de artă expuse de acesta în muzeu.  

Să se definească un declanşator prin care reactualizările executate asupra 

vizualizării vi_op_ar se vor transmite automat tabelelor nou_opera şi nou_artist. 

CREATE TABLE nou_opera AS 

  SELECT cod_opera, cod_artist, valoare, tip, stil 

  FROM   opera; 

CREATE TABLE nou_artist AS 

  SELECT cod_artist, nume, sum_val 

  FROM   artist; 

CREATE VIEW vi_op_ar AS 

  SELECT cod_opera,o.cod_artist,valoare,tip,nume, 

         sum_val 

  FROM   opera o, artist a 

  WHERE  o.cod_artist = a.cod_artist 

CREATE OR REPLACE TRIGGER react 

INSTEAD OF INSERT OR DELETE OR UPDATE ON vi_op_ar 

FOR EACH ROW 

BEGIN 

  IF INSERTING THEN 

    INSERT INTO nou_opera 

    VALUES (:NEW.cod_opera, :NEW.cod_artist, :NEW.valoare, 

            :NEW.tip); 

    UPDATE nou_artist 

    SET    sum_val = sum_val + :NEW.valoare 

    WHERE  cod_artist = :NEW.cod_artist; 

  ELSIF DELETING THEN 

    DELETE FROM nou_opera 

    WHERE  cod_opera = :OLD.cod_opera; 

    UPDATE nou_artist 

    SET    sum_val = sum_val - :OLD.valoare 

    WHERE  cod_artist = :OLD.cod_artist; 

  ELSIF UPDATING ('valoare') THEN 

    UPDATE nou_opera 

    SET    valoare = :NEW.valoare 

    WHERE  cod_opera = :OLD.cod_opera; 

    UPDATE nou_artist 

    SET   sum_val = sum_val + (:NEW.valoare - :OLD.valoare) 

    WHERE  cod_artist = :OLD.cod_artist; 

  ELSIF UPDATING ('cod_artist') THEN 

    UPDATE nou_opera 

    SET    cod_artist = :NEW.cod_artist 

    WHERE  cod_opera = :OLD.cod_opera; 

    UPDATE nou_artist 

    SET    sum_val = sum_val - :OLD.valoare 

    WHERE  cod_artist = :OLD.cod_artist; 

    UPDATE nou_artist 

    SET    sum_val = sum_val + :NEW.valoare 

    WHERE  cod_artist = :NEW.cod_artist; 

  END IF; 

END; 

/ 

Declanşatori sistem 

Declanşatorii sistem sunt activaţi de comenzi LDD (CREATE, DROP, 

ALTER) şi de anumite evenimente sistem (STARTUP, SHUTDOWN, LOGON, 

LOGOFF, SERVERERROR, SUSPEND). Un declanşator sistem poate fi definit la 

nivelul bazei de date sau la nivelul schemei.  

Sintaxa pentru crearea unui astfel de declanşator este următoarea: 

CREATE [OR REPLACE] TRIGGER [schema.]nume_declanşator 

  {BEFORE | AFTER} 

  {lista_evenimente_LDD | lista_evenimente_bază} 

  ON {DATABASE | SCHEMA} 

  [WHEN (condiţie) ] 

  corp_declanşator; 

Cuvintele cheie DATABASE sau SCHEMA specifică nivelul 

declanşatorului. 

Există restricţii asupra expresiilor din condiţia clauzei WHEN. De exemplu, 

declanşatorii LOGON şi LOGOFF pot verifica doar identificatorul (userid) şi 

numele utilizatorului (username), iar declanşatorii LDD pot verifica tipul şi 

numele obiectelor definite, identificatorul şi numele utilizatorului. 

Evenimentele amintite anterior pot fi asociate clauzelor BEFORE sau 

AFTER. De exemplu, un declanşator LOGON (AFTER) se activează după ce un 

utilizator s-a conectat la baza de date, un declanşator CREATE (BEFORE sau 

AFTER) se activează înainte sau după ce a fost creat un obiect al bazei, un 

declanşator SERVERERROR (AFTER) se activează ori de câte ori apare o eroare 

(cu excepţia erorilor: ORA-01403, ORA-01422, ORA-01423, ORA-01034, ORA-

04030). 

Declanşatorii LDD se activează numai dacă obiectul creat este de tip table, 

cluster, function, procedure, index, package, role, sequence, synonym, 

tablespace, trigger, type, view sau user. 

Pentru declanşatorii sistem se pot utiliza funcţii speciale care permit 

obţinerea de informaţii referitoare la evenimentul declanşator. Ele sunt funcţii 

PL/SQL stocate care trebuie prefixate de numele proprietarului (SYS). 

Printre cele mai importante funcţii care furnizează informaţii referitoare la 

evenimentul declanşator, se remarcă: 

 SYSEVENT – returnează evenimentul sistem care a activat declanşatorul 

(este de tip VARCHAR2(20) şi este aplicabilă oricărui eveniment); 

 DATABASE_NAME – returnează numele bazei de date curente (este de 

tip VARCHAR2(50) şi este aplicabilă oricărui eveniment); 

 SERVER_ERROR – returnează codul erorii a cărei poziţie în stiva 

erorilor este dată de argumentul de tip NUMBER al funcţiei (este de tip 

NUMBER şi este aplicabilă evenimentului SERVERERROR); 

 LOGIN_USER – returnează identificatorul utilizatorului care activează 

declanşatorul (este de tip VARCHAR2(30) şi este aplicabilă oricărui 

eveniment); 

 DICTIONARY_OBJ_NAME – returnează numele obiectului la care face 

referinţă comanda LDD ce a activat declanşatorul (este de tip 

VARCHAR2(30) şi este aplicabilă evenimentelor CREATE, ALTER, 

DROP). 

Exemplu: 

CREATE OR REPLACE TRIGGER logutiliz 

  AFTER CREATE ON SCHEMA 

BEGIN 

  INSERT INTO ldd_tab(user_id, object_name, creation_date) 

  VALUES      (USER, SYS.DICTIONARY_OBJ_NAME, SYSDATE); 

END logutiliz; 

Evenimentul SERVERERROR poate fi utilizat pentru a urmări erorile care 

apar în baza de date. Codul erorii este furnizat, prin intermediul declanşatorului, 

de funcţia SERVER_ERROR, iar mesajul asociat erorii poate fi obţinut cu 

procedura DBMS_UTILITY.FORMAT_ERROR_STACK. 

Exemplu: 

CREATE TABLE erori ( 

  moment       DATE, 

  utilizator   VARCHAR2(30), 

  nume_baza    VARCHAR2(50), 

  stiva_erori  VARCHAR2(2000) ); 

/ 

CREATE OR REPLACE TRIGGER logerori 

  AFTER SERVERERROR ON DATABASE 

BEGIN 

  INSERT INTO erori 

  VALUES (SYSDATE, SYS.LOGIN_USER, SYS.DATABASE_NAME, 

          DBMS_UTILITY.FORMAT_ERROR_STACK); 

END logerori; 

/ 

Modificarea şi suprimarea declanşatorilor 

Opţiunea OR REPLACE din cadrul comenzii CREATE TRIGGER 

recreează declanşatorul, dacă acesta există. Clauza permite schimbarea definiţiei 

unui declanşator existent fără suprimarea acestuia. 

Similar procedurilor şi pachetelor, un declanşator poate fi suprimat prin: 

DROP TRIGGER [schema.]nume_declanşator; 

Uneori acţiunea de suprimare a unui declanşator este prea drastică şi este 

preferabilă doar dezactivarea sa temporară. În acest caz, declanşatorul va 

continua să existe în dicţionarul datelor.  

Modificarea unui declanşator poate consta din recompilarea (COMPILE), 

redenumirea (RENAME), activarea (ENABLE) sau dezactivarea (DISABLE) 

acestuia şi se realizează prin comanda: 

ALTER TRIGGER [schema.]nume declanşator  

  {ENABLE | DISABLE | COMPILE | RENAME TO nume_nou}  

  {ALL TRIGGERS} 

Dacă un declanşator este activat, atunci sistemul Oracle îl execută ori de 

câte ori au loc operaţiile precizate în declanşator asupra tabelului asociat şi când 

condiţia de restricţie este îndeplinită. Dacă declanşatorul este dezactivat, atunci 

sistemul Oracle nu îl va mai executa. După cum s-a mai subliniat, dezactivarea 

unui declanşator nu implică ştergerea acestuia din dicţionarul datelor.  

Toţi declanşatorii asociaţi unui tabel pot fi activaţi sau dezactivaţi utilizând 

opţiunea ALL TRIGGERS (ENABLE ALL TRIGGERS, respectiv DISABLE ALL 

TRIGGERS). Declanşatorii sunt activaţi în mod implicit atunci când sunt creaţi. 

Pentru activarea (enable) unui declansator, server-ul Oracle:  

 verifica integritatea constrangerilor,  

 garanteaza ca declansatorii nu pot compromite constrangerile de 

integritate,  

 garanteaza consistenta la citire a vizualizarilor,  

 gestioneaza dependentele. 

Activarea şi dezactivarea declanşatorilor asociaţi unui tabel se poate 

realiza şi cu ajutorul comenzii ALTER TABLE. 

Un declanşator este compilat în mod automat la creare. Dacă un site este 

neutilizabil atunci când declanşatorul trebuie compilat, sistemul Oracle nu poate 

valida comanda de accesare a bazei distante şi compilarea eşuează. 

Informaţii despre declanşatori 

În DD există vizualizări ce conţin informaţii despre declanşatori şi despre 

starea acestora (USER_TRIGGERS, USER_TRIGGER_COL, ALL_TRIGGERS, 

DBA_TRIGGERS etc.). Aceste vizualizări sunt actualizate ori de câte ori un 

declanşator este creat sau suprimat.  

Atunci când declanşatorul este creat, codul său sursă este stocat în 

vizualizarea USER_TRIGGERS. Vizualizarea ALL_TRIGGERS conţine informaţii 

despre toţi declanşatorii din baza de date. Pentru a detecta dependenţele 

declanşatorilor poate fi consultată vizualizarea USER_DEPENDENCIES, iar 

ALL_DEPENDECIES conţine informaţii despre dependenţele tuturor obiectelor 

din baza de date. Erorile rezultate din compilarea declanşatorilor pot fi analizate 

din vizualizarea USER_ERRORS, iar prin comanda SHOW ERRORS se vor afişa 

erorile corespunzătoare ultimului declanşator compilat. 

În operaţiile de gestiune a bazei de date este necesară uneori reconstruirea 

instrucţiunilor CREATE TRIGGER, atunci când codul sursă original nu mai este 

disponibil. Aceasta se poate realiza utilizând vizualizarea USER_TRIGGERS.  

Vizualizarea include numele declanşatorului (TRIGGER_NAME), tipul 

acestuia (TRIGGER_TYPE), evenimentul declanşator (TRIGGERING_EVENT), 

numele proprietarului tabelului (TABLE_OWNER), numele tabelului pe care este 

definit declanşatorul (TABLE_NAME), clauza WHEN (WHEN_CLAUSE), corpul 

declanşatorului (TRIGGER_BODY), antetul (DESCRIPTION), starea acestuia 

(STATUS) care poate să fie ENABLED sau DISABLED şi numele utilizate pentru 

a referi parametrii OLD şi NEW (REFERENCING_NAMES). Dacă obiectul de 

bază nu este un tabel sau o vizualizare, atunci TABLE_NAME este null. 

Exemplu: 

Presupunând că nu este disponibil codul sursă pentru declanşatorul alfa, să 

se reconstruiască instrucţiunea CREATE TRIGGER corespunzătoare acestuia. 

SELECT  ’CREATE OR REPLACE TRIGGER ’ || DESCRIPTION || 

        TRIGGER_BODY 

FROM    USER_TRIGGERS 

WHERE   TRIGGER_NAME = 'ALFA'; 

Cu această interogare se pot reconstrui numai declanşatorii care aparţin 

contului utilizator curent. O interogare a vizualizărilor ALL_TRIGGERS sau 

DBA_TRIGGERS permite reconstruirea tuturor declanşatorilor din sistem, dacă se 

dispune de privilegii DBA. 

Exemplu: 

SELECT  USERNAME 

FROM    USER_USERS; 

Aceasta cerere furnizeaza numele "proprietarului" (creatorului) 

declansatorului si nu numele utilizatorului care a reactualizat tabelul. 

Privilegii sistem 

Sistemul furnizează privilegii sistem pentru gestiunea declanşatorilor:  

 CREATE TRIGGER (permite crearea declanşatorilor în schema 

personală); 

 CREATE ANY TRIGGER (permite crearea declanşatorilor în orice 

schemă cu excepţia celei corespunzătoare lui SYS); 

 ALTER ANY TRIGGER (permite activarea, dezactivarea sau compilarea 

declanşatorilor în orice schemă cu excepţia lui SYS); 

 DROP ANY TRIGGER (permite suprimarea declanşatorilor la nivel de 

bază de date în orice schemă cu excepţia celei corespunzătoate lui SYS); 

 ADMINISTER DATABASE TRIGGER (permite crearea sau modificarea 

unui declanşator sistem referitor la baza de date); 

 EXECUTE (permite referirea, în corpul declanşatorului, a procedurilor, 

funcţiilor sau pachetelor din alte scheme).  

Tabele mutating 

Asupra tabelelor şi coloanelor care pot fi accesate de corpul 

declanşatorului există anumite restricţii. Pentru a analiza aceste restricţii este 

necesară definirea tabelelor în schimbare (mutating) şi constrânse (constraining). 

Un tabel constraining este un tabel pe care evenimentul declanşator trebuie 

să-l consulte fie direct, printr-o instrucţiune SQL, fie indirect, printr-o 

constrângere de integritate referenţială declarată. Tabelele nu sunt considerate 

constraining în cazul declanşatorilor la nivel de instrucţiune. Comenzile SQL din 

corpul unui declanşator nu pot modifica valorile coloanelor care sunt declarate 

chei primare, externe sau unice (PRIMARY KEY, FOREIGN KEY, UNIQUE 

KEY) într-un tabel constraining. 

Un tabel mutating este tabelul modificat de instrucţiunea UPDATE, 

DELETE sau INSERT, sau un tabel care va fi actualizat prin efectele acţiunii 

integrităţii referenţiale ON DELETE CASCADE. Chiar tabelul pe care este definit 

declanşatorul este un tabel mutating, ca şi orice tabel referit printr-o constrângere 

FOREING KEY. Tabelele nu sunt considerate mutating pentru declanşatorii la 

nivel de instrucţiune, cu excepţia celor declanşaţi ca efect al opţiunii ON 

DELETE CASCADE. Vizualizările nu sunt considerate mutating în declanşatorii 

INSTEAD OF. 

Regula care trebuie respectată la utilizarea declanşatoriilor este: 

comenzile SQL din corpul unui declanşator nu pot consulta sau modifica 

date dintr-un tabel mutating. 

Exceptia! Dacă o comandă INSERT afectează numai o înregistrare, 

declanşatorii la nivel de linie (BEFORE sau AFTER) pentru înregistrarea 

respectivă nu tratează tabelul ca fiind mutating. Acesta este unicul caz în care un 

declanşator la nivel de linie poate citi sau modifica tabelul. Comanda INSERT 

INTO tabel SELECT … consideră tabelul mutating chiar dacă cererea returnează 

o singură linie. 

Exemplu: 

CREATE OR REPLACE TRIGGER cascada 

  AFTER UPDATE OF cod_artist ON artist 

  FOR EACH ROW 

BEGIN 

  UPDATE opera 

  SET    opera.cod_artist = :NEW.cod_artist 

  WHERE  opera.cod_artist = :OLD.cod_artist 

END; 

UPDATE   artist 

SET      cod_artist = 71 

WHERE    cod_artist = 23; 

La execuţia acestei secvenţe este semnalată o eroare. Tabelul artist 

referenţiază tabelul opera printr-o constrângere de cheie externă. Prin urmare, 

tabelul opera este constraining, iar declanşatorul cascada încearcă să schimbe 

date în tabelul constraining, ceea ce nu este permis. Exemplul va funcţiona corect 

dacă nu este definită sau activată constrângerea referenţială între cele două tabele. 

Exemplu: 

Să se implementeze cu ajutorul unui declanşator restricţia că într-o sală pot 

să fie expuse maximum 10 opere de artă. 

CREATE OR REPLACE TRIGGER TrLimitaopere 

  BEFORE INSERT OR UPDATE OF cod_sala ON opera 

  FOR EACH ROW 

DECLARE  

  v_Max_opere CONSTANT NUMBER := 10; 

  v_opere_curente      NUMBER; 

BEGIN 

  SELECT  COUNT(*) INTO v_opere_curente 

  FROM    opera 

  WHERE   cod_sala = :NEW.cod_sala; 

  IF v_opere_curente + 1 > v_Max_opere THEN 

    RAISE_APPLICATION_ERROR(-20000,'Prea multe opere de  

      arta in sala avand codul ' || :NEW.cod_sala); 

  END IF; 

END TrLimitaopere; 

Cu toate că declanşatorul pare să producă lucrul dorit, totuşi după o 

reactualizare a tabelului opera în următoarea manieră: 

INSERT INTO opera (cod_opera, cod_sala)  

VALUES (756893, 10); 

se obţine următorul mesaj de eroare: 

ORA-04091: tabel opera is mutating, trigger/function 

may not see it 

ORA-04088: error during execution of trigger 

Eroarea ORA-04091 apare deorece declanşatorul TrLimitaopere consultă 

chiar tabelul (opera) la care este asociat declanşatorul (mutating). 

Tabelul opera este mutating doar pentru un declanşator la nivel de linie. 

Aceasta înseamnă că tabelul poate fi consultat în interiorul unui declanşator la 

nivel de instrucţiune. Totuşi, limitarea numărului operelor de artă nu poate fi 

făcută în interiorul unui declanşator la nivel de instrucţiune, din moment ce este 

necesară valoarea :NEW.cod_sala în corpul declanşatorului.  

CREATE OR REPLACE PACKAGE PSalaDate AS 

  TYPE t_cod_sala  IS TABLE OF opera.cod_sala%TYPE 

                INDEX BY BINARY_INTEGER; 

  TYPE t_cod_opera IS TABLE OF opera.cod_opera%TYPE 

                INDEX BY BINARY_INTEGER; 

  v_cod_sala    t_cod_sala; 

  v_cod_opera   t_cod_opera; 

  v_NrIntrari  BINARY_INTEGER := 0; 

END PSalaDate; 

CREATE OR REPLACE TRIGGER TrLLimitaSala 

  BEFORE INSERT OR UPDATE OF cod_sala ON opera 

  FOR EACH ROW 

BEGIN 

  PSalaDate.v_NrIntrari := PSalaDate.v_NrIntrari + 1; 

  PSalaDate.v_cod_sala(PSalaDate.v_NrIntrari) := 

                                    :NEW.cod_sala; 

  PSalaDate.v_cod_opera(PSalaDate.v_NrIntrari) := 

                                    :NEW.cod_opera; 

END TrLLimitasala; 

CREATE OR REPLACE TRIGGER TrILimitaopere 

  AFTER INSERT OR UPDATE OF cod_sala ON opera 

DECLARE  

  v_Max_opere      CONSTANT NUMBER := 10; 

  v_opere_curente  NUMBER; 

  v_cod_operax     opera.cod_opera%TYPE; 

  v_cod_salax      opera.cod_sala%TYPE; 

BEGIN 

  FOR v_LoopIndex IN 1..PsalaDate.v_NrIntrari LOOP 

    v_cod_operax := PsalaDate.v_cod_opera(v_LoopIndex); 

    v_cod_salax := PsalaDate.v_cod_sala(v_LoopIndex); 

    SELECT  COUNT(*) 

    INTO    v_opere_curente 

    FROM    opera 

    WHERE   cod_sala = v_cod_salax; 

    IF v_opere_curente > v_Max_opere THEN 

      RAISE_APPLICATION_ERROR(-20000, 'Prea multe opere de 

      arta in sala' || v_cod_salax || 'din cauza inserarii 

      operei avand codul' || v_cod_operax) 

    END IF; 

  END LOOP; 

  /* Reseteaza contorul deoarece urmatoarea executie  

     va folosi date noi */ 

  PSalaDate.v_NrIntrari := 0; 

END TrILimitaopere; 

O solutie pentru acestă problemă este crearea a doi declanşatori, unul la 

nivel de linie şi altul la nivel de instrucţiune. În declanşatorul la nivel de linie se 

înregistrează valoarea lui :NEW.cod_opera, dar nu va fi interogat tabelul opera.  

Interogarea va fi făcută în declanşatorul la nivel de instrucţiune şi va folosi 

valoarea înregistrată în declanşatorul la nivel de linie. 

O modalitate pentru a înregistra valoarea lui :NEW.cod_opera este 

utilizarea unui tablou indexat în interiorul unui pachet. 

Exemplu: 

Să se creeze un declanşator care: 

a) dacă este eliminată o sală, va şterge toate operele expuse în sala 

respectivă; 

b) dacă se schimbă codul unei săli, va modifica această valoare pentru 

fiecare operă de artă expusă în sala respectivă. 

CREATE OR REPLACE TRIGGER sala_cascada 

  BEFORE DELETE OR UPDATE OF cod_sala ON sala 

  FOR EACH ROW 

BEGIN 

  IF DELETING THEN 

    DELETE FROM opera 

    WHERE  cod_sala = :OLD.cod_sala; 

  END IF; 

  IF UPDATING AND :OLD.cod_sala != :NEW.cod_sala THEN 

    UPDATE opera 

    SET    cod_sala = :NEW.cod_sala 

    WHERE  cod_sala = :OLD.cod_sala; 

  END IF; 

END sala_cascada; 

Declanşatorul anterior realizează constrângerea de integritate UPDATE sau 

ON DELETE CASCADE, adică ştergerea sau modificarea cheii primare a unui 

tabel „părinte“ se va reflecta şi asupra înregistrărilor corespunzătoare din tabelul 

„copil“. 

Executarea acestuia, pe tabelul sala (tabelul „părinte“), va duce la 

efectuarea a două tipuri de operaţii pe tabelul opera (tabelul „copil“). 

La eliminarea unei săli din tabelul sala, se vor şterge toate operele de artă 

corespunzătoare acestei săli. 

DELETE FROM  sala  

WHERE  cod_sala = 773; 

La modificarea codului unei săli din tabelul sala, se va actualiza codul sălii 

atât în tabelul sala, cât şi în tabelul opera. 

UPDATE  sala  

SET     cod_sala = 777  

WHERE   cod_sala = 333; 

Se presupune că asupra tabelului opera există o constrângere de integritate: 

FOREIGN KEY (cod_sala) REFERENCES sala(cod_sala) 

În acest caz sistemul Oracle va afişa un mesaj de eroare prin care se 

precizează că tabelul sala este mutating, iar constrângerea definită mai sus nu 

poate fi verificată. 

ORA-04091: table MASTER.SALA is mutating,  

           trigger/function may not see it 

Pachetele pot fi folosite pentru încapsularea detaliilor logice legate de 

declanşatori. Exemplul următor arată un mod simplu de implementare a acestei 

posibilităţi. Este permisă apelarea unei proceduri sau funcţii stocate din blocul 

PL/SQL care reprezintă corpul declanşatorului. 

Exemplu: 

CREATE OR REPLACE PACKAGE pachet IS 

  PROCEDURE procesare_trigger(pvaloare IN NUMBER, 

                              pstare   IN VARCHAR2); 

END pachet; 

CREATE OR REPLACE PACKAGE BODY pachet IS 

  PROCEDURE procesare_trigger(pvaloare IN NUMBER, 

                              pstare   IN VARCHAR2) IS 

  BEGIN 

  … 

  END procesare_trigger; 

END pachet; 

CREATE OR REPLACE TRIGGER gama 

  AFTER INSERT ON opera 

  FOR EACH ROW 

BEGIN 

  pachet.procesare_trigger(:NEW.valoare,:NEW.stare) 

END; 

 


